

CustomTicks package

Mark A. Caprio, Department of Physics, University of Notre Dame

Version 1.70 (June 3, 2011)

Introduction

Mathematica provides a powerful system for generating graphics but does not provide, in built-in form, the fine formatting control necessary for the preparation of publication quality figures. The CustomTicks package provides detailed customization of tick mark placement and formatting. The flexibility achieved matches or exceeds that available with most commercial scientific plotting software. Linear, logarithmic, and general nonlinear axes are supported. Some tick mark manipulation functions, for use in graphics programming, are also provided by the CustomTicks package.

The CustomTicks package was developed as part of the LevelScheme system for preparing publication-quality scientific figures [Comput. Phys. Commun. **171**, 107 (2005)].

Basic use for linear axes

The default tick marks produced by *Mathematica*'s plotting functions are typically not ideal for publication. It is often desirable to be able to change the tick spacing from that selected by *Mathematica*. The tick marks are also often too short to be easily visible.

It is possible to override the default *Mathematica* ticks by specifying a list of tick marks, complete with formatting information, as the value for the `Ticks` or `FrameTicks` option (see the *Mathematica* documentation for basic plotting options). It is prohibitively tedious to construct such lists by hand. The CustomTicks package provides functions to automatically construct lists of tick marks, with detailed control over formatting.


<code>LinTicks[<i>x1</i>, <i>x2</i>]</code>	Produces linear tick specifications, with automatically chosen major and minor tick intervals
<code>LinTicks[<i>x1</i>, <i>x2</i>, <i>interval</i>, <i>subdivs</i>]</code>	Produces linear tick specifications, with manually chosen major and minor tick intervals

Tick specification function.

option name	default value	
TickRange	{-Infinity, Infinity}	Limits the drawing of ticks (and their labels) to given coordinate range
ShowTickLabels	True	Controls whether or not major tick labels are printed
TickLabelRange	{-Infinity, Infinity}	Limits printing of major tick labels to given coordinate range
ShowFirst	True	Controls whether or not first major tick label is printed
ShowLast	True	Controls whether or not last major tick label is printed
TickLabelStep	1	Limits printing of major tick labels to one in every TickLabelStep major ticks
TickLabelStart	0	Used in conjunction with TickLabelStep chooses which subset of major tick labels are printed
MajorTickLength	{0.01, 0}	List specifying the lengths for the major ticks (into and out of the frame, as described in the <i>Mathematica</i> documentation for Ticks)
MinorTickLength	{0.005, 0}	List specifying the lengths for the minor ticks (into and out of the frame, as described in the <i>Mathematica</i> documentation for Ticks)
MajorTickStyle	{}	List specifying the line style for the major ticks
MinorTickStyle	{}	List specifying the line style for the minor ticks
DecimalDigits	Automatic	Sets number of digits after decimal place for major tick labels; if Automatic, the maximum number of digits needed for any label is used
ShowMinorTicks	True	Controls whether or not the minor ticks are drawn; mainly for use with LogTicks (see below)
ExtraTicks	{}	Additional coordinate values at which to insert tick marks

Tick formatting options.

The `LinTicks` function, in its simplest form, is given a starting and an ending coordinate value as its arguments. It then generates the same tick marks *Mathematica* would have for this coordinate range, except that the tick marks lines are somewhat longer. Alternatively, arguments may be given to manually specify the coordinate interval between major tick marks and the number of minor subdivisions. Several further options, listed above, can be specified. These control which tick marks are drawn, which major ticks have labels, and the formatting of the tick lines and labels. Some examples are shown below.



In Mathematica versions 5 and earlier, Mathematica dropped trailing zeros after the decimal point in its default tick marks, leading to a series of ticks of “ragged” lengths (e.g., “0.”, “0.5”, “1.”, ...), which were particularly unsuited for publication. The CustomTicks package therefore provided not simply finer control and customization but also much “neater” formatting than the Mathematica default.

If you are using Mathematica 5 or an earlier version, the rightmost example assumes the Mathematica color names package (`Graphics`Colors``) has been loaded.

The tick specifications generated with the CustomTicks package can be used either with two-dimensional plots (through the options `Ticks` or `FrameTicks`) or with three-dimensional plots (through the option `Ticks`). These options are described further in the basic *Mathematica* documentation on plotting.


Logarithmic axes

The function `LogTicks` generates tick marks for logarithmic axes. `LogTicks` can produce tick marks for an arbitrary logarithmic base (10 is the default, but e and 2 are other commonly useful bases). Unlike the *Mathematica* `LogPlot` function, which produces cumbersome decimal labels (e.g., “0.0000001”, “0.000001”, ...), `LogTicks` produces true exponential labels (e.g., “ 10^{-7} ”, “ 10^{-6} ”, ...).

<code>LogTicks[n1, n2]</code>	Produces logarithmic tick specifications, base 10
<code>LogTicks[base, n1, n2]</code>	Produces logarithmic tick specifications, arbitrary <i>base</i>


Logarithmic tick specification function.

`LogTicks` must be given the starting power, ending power, and, optionally, the logarithmic base b . For base 10, a total of eight minor ticks are produced in each major interval, at 2×10^n through 9×10^n . For an arbitrary base b , $\lceil b \rceil - 2$ minor ticks are produced, at $2 \times b^n$, $3 \times b^n$, Display of the minor ticks may be suppressed by specifying the option `ShowMinorTicks → False`. Some examples are shown below.




Base 10

```
LogTicks[0, 2]
```




Skipped labels

```
LogTicks[-8, 8,  
TickLabelStep -> 4]
```



No minor ticks

```
LogTicks[-8, 8,  
ShowMinorTicks -> False,  
TickLabelStep -> 4]
```

Base e


```
LogTicks[E, 0, 4]
```

Note that plots with *logarithmic* axes are actually generated as *linear* plots, but where the logarithm has been taken of either the x-axis or y-axis variable. Specifically, for base 10,


- (1) a *logarithmic* (or *linear-log*) plot of f is obtained by plotting $\log_{10} f(x)$,
- (2) a *log-linear* plot of f is obtained by plotting $f(10^x)$, and
- (3) a *log-log* plot of f is obtained by plotting $\log_{10} f(10^x)$

on ordinary linear axes. A similar procedure holds for bases other than 10. Examples of a logarithmic plot and a log-log plot follow.

```
Plot[  
  {Log[10, Cosh[x]], Log[10, Sinh[x]]}, {x, 0, 10},  
  PlotRange -> {{-0.0001, 4}, {-0.5, 2.5}},  
  FrameTicks -> {  
    LinTicks[0, 4],  
    LogTicks[10, -1, 3],  
    LinTicks[0, 4, ShowTickLabels -> False],  
    LogTicks[10, -1, 3, ShowTickLabels -> False]  
  },  
  Axes -> False, Frame -> True, ImageSize -> 72*3  
]
```




```
Plot[
  {Log[10, (10^x)^2], Log[10, (10^x)^5]}, {x, -1, 3},
  PlotRange -> {{-0.0001, 3}, {-0.5, 4.5}},
  FrameTicks -> {
    LogTicks[10, 0, 3],
    LogTicks[10, -1, 5],
    None,
    None
  },
  Axes -> False, Frame -> True, ImageSize -> 72*3
]
```



Automatic ticks for *Mathematica* plot functions


The functions `LinTicks` and `LogTicks` can also be specified as automatic tick generation functions for the *Mathematica* plotting functions. This saves you typing the plot range explicitly each time, at least if you do not wish to specify details such as the number of subdivisions.

```
Plot[
  {Log[10, Cosh[x]], Log[10, Sinh[x]]}, {x, 0, 10},
  PlotRange -> {{-0.0001, 4}, {-0.5, 2.5}},
  FrameTicks -> {LinTicks, LogTicks, None, None},
  Axes -> False, Frame -> True, ImageSize -> 72*3
]
```




What if you wish to have tick marks on the top and right as well, but without labels on them? Simply specifying `FrameTicks -> {LinTicks, LogTicks}` or `FrameTicks -> {LinTicks, LogTicks, LinTicks, LogTicks}` would unfortunately result in ticks with unsightly and redundant labels. This can be avoided with the `CustomTicks StripTickLabels` function, as shown below.

```
Plot[
  {Log[10, Cosh[x]], Log[10, Sinh[x]]}, {x, 0, 10},
  PlotRange -> {{-0.0001, 4}, {-0.5, 2.5}},
  FrameTicks ->
  {LinTicks, LogTicks, StripTickLabels[LinTicks], StripTickLabels[LogTicks]},
  Axes -> False, Frame -> True, ImageSize -> 72*3
]
```



If you are doing many such plots, it is easiest to set the necessary options as the default options for `Plot`.

```
SetOptions[Plot, Axes->False, Frame->True, FrameTicks->
{LinTicks, LogTicks, StripTickLabels[LinTicks], StripTickLabels[LogTicks] }];
GraphicsGrid[{{Plot[Log[10, x^2], {x, 0, 10}], Plot[Log[10, x^-2], {x, 0, 10}]}]}
```




option name	default value	
LogPlot	False	Controls whether LogTicks operates as it should for standalone use or for automatic tick mark generation with <i>Mathematica</i> 's LogPlot

LogTicks special option.

If you wish to use LogPlot or related *Mathematica* functions directly for your logarithmic plots, you may also do so. However, LogTicks must then be instructed to adjust its output accordingly. (LogPlot expects tick coordinates to be specified as the true coordinate value, not the logarithm of the coordinate value. LogTicks must also interpret its *n1* and *n2* arguments differently in this case.) Automatic use with LogPlot can be accomplished by setting the option LogPlot->True.

```
SetOptions[LogTicks, LogPlot -> True];
Plot[
{Cosh[x], Sinh[x]}, {x, 0, 10},
PlotRange -> {{-0.0001, 4}, {10^-0.5, 10^2.5}},
FrameTicks -> {LinTicks, LogTicks, StripTickLabels[LinTicks],
StripTickLabels[LogTicks]},
Frame -> True, ImageSize -> 72*3
]
```




Advanced customization

option name	default value	
TickLabelFunction	Automatic	Function used to generate major tick labels (first argument is the numerical coordinate, second argument is the <code>LinTicks</code> default formatted label); <code>Automatic</code> gives the default label
TickPreTransformation	Identity	Function to be applied to tick coordinates, before range tests and label generation
TickPostTransformation	Identity	Function to be applied to tick coordinates, after range tests and label generation
MinorTickIndexTransformation	Identity	Function to be applied to minor tick indices (originally 1, 2, ..., <i>subdivs</i> - 1) before minor tick coordinate is obtained by linear interpolation between major tick positions
MinorTickIndexRange	{1, Infinity}	Limits drawing of minor ticks to those with indices (before transformation) in given range

Advanced customization options.

`LinTicks` accepts several options for advanced customization, allowing fully customizable labels and general nonlinear axis scales. The option `TickLabelFunction` is used to specify the function to be used to construct tick labels (see the *Mathematica* documentation for `Function` for information on defining functions). The label function is given as arguments both the raw numerical tick coordinate and the `LinTicks` default formatted label, so it can work with whichever is more convenient. The label function may be used for simple tasks, such as attaching a prefix or suffix to the usual default label, or for more sophisticated formatting. In the following example, tick values are formatted as rational multiples of π .

```
LinTicks[0, 2*Pi, Pi/2, 4, TickLabelFunction -> (Rationalize[#/Pi]*Pi &)]
```


Nonlinear axes are constructed using the coordinate transformation functions. For instance, the `LogTicks` function provided by the CustomTicks package is actually implemented as a special case of `LinTicks`, with transformed minor tick positions. A simplified version (base 10 logarithm only) is given below for illustration.

```
Log10Ticks[p1_Integer, p2_Integer, Opts___?OptionQ] := LinTicks[
  p1, p2, 1, 9,
  TickLabelFunction -> (DisplayForm[SuperscriptBox[10, IntegerPart[#]]] &),
  MinorTickIndexTransformation -> (Log[10, # + 1] * 9 &),
  Opts
];
```

<code>LinTicks[<i>majorticklist</i>, <i>minorticklist</i>]</code>	Produces major and minor ticks at the specified coordinate values
---	---

Form of tick specification function for ticks at arbitrary locations.

Ticks may be placed at arbitrary coordinate locations by using the most flexible form of `LinTicks`, in which all major and minor tick coordinates are specified explicitly in two lists. All the usual coordinate-transformation and customization options described above (except `MinorTickIndexRange`) are still applicable.



Manual choice of tick coordinates

```
LinTicks[{0, 1, 2, 3, 5}, Range[0.1, 0.9, 0.1]]
```

Tick mark programming utilities

<code>LimitTickRange[{x1, x2}, ticks]</code>	Selects those tick mark with coordinates in the range specified; approximate equality testing is used to avoid dropping ticks at the ends of the interval due to roundoff; ticks must be specified as lists rather than bare numbers
<code>TransformTicks[coordfcn, lengthfcn, ticks]</code>	Applies the specified transformation functions to the tick coordinates and tick lengths, respectively; ticks must be specified with an explicit pair of in and out lengths
<code>StripTickLabels[ticks]</code>	Removes any text labels from ticks; ticks must be specified as lists rather than bare numbers
<code>AugmentTicks[labelfcn, {l1, l2}, stylelist, ticks]</code>	Upgrades all tick specifications to full specifications, complete with labels, lengths into and out of the frame (default 0 for out), and style directives
<code>AugmentAxisTickOptions[numaxes, tickoptions]</code>	Given a list of tick options (themselves lists of tick specifications) for several axes, replaces any <code>None</code> entries with null lists and appends additional null lists as needed to make <i>numaxes</i> entries; a value <code>None</code> for <i>tickoptions</i> is replaced by a list of null lists
<code>TickQ[x]</code>	Tests whether or not <i>x</i> is a valid tick mark specification
<code>TickListQ[x]</code>	Tests whether or not <i>x</i> is a list of valid tick mark specifications

Tick manipulation utilities.

Several utility functions for tick mark manipulation and testing are provided. These are mainly intended for use in graphics programming rather than for direct use by someone wishing to specify tick marks. They are used internally by the `LevelScheme` figure preparation system.

<code>FractionDigits[x]</code>	Returns the number of digits to the right of the point in the decimal representation of <i>x</i>
--------------------------------	--

Decimal digit counting function.

`FractionDigits` determines the number of digits to the right of the point in the decimal representation of a number. It is of use in constructing fixed-point tick labels. It will, naturally, return large values, determined by *Mathematica*'s `Precision`, for some numbers, such as non-terminating rationals. It accepts the option `FractionDigitsBase`, by default 10, for work with non-decimal representations. Some examples follow:

```

FractionDigits[100]
FractionDigits[1.25]
FractionDigits[1 / 3]

0
2
17

```

© Copyright 2011, Mark A. Caprio.